
OBJECT Codeunit 92600 SpecDimensionManagement

{

  OBJECT-PROPERTIES

  {

    Date=19/12/17;

    Time=12:00:00;

    Modified=Yes;

    Version List=SDM;

  }

  PROPERTIES

  {

    OnRun=BEGIN

          END;



  }

  CODE

  {

    VAR

      TempDimensionSetEntry@1100074000 : TEMPORARY Record 480;



    PROCEDURE INIT@39(_RelatedToRecord@1100074004 : Variant);

    VAR

      DataTypeManagement@1100074005 : Codeunit 701;

      RecRef@1100074007 : RecordRef;

      FldRef@1100074008 : FieldRef;

      DimCode@1100074002 : ARRAY [20] OF Code[20];

      DimValueCode@1100074001 : ARRAY [20] OF Code[20];

      OldDimSetID@1100074003 : Integer;

      i@1008 : Integer;

    BEGIN

      // Field 480 is "Dimension Set ID" in standard tables

      IF NOT DataTypeManagement.GetRecordRefAndFieldRef(_RelatedToRecord, 480, RecRef, FldRef) THEN

        EXIT;



      EVALUATE(OldDimSetID, FORMAT(FldRef.VALUE));

      GetGLSetupShortcutDimCodes(DimCode);

      GetCurrDimValues(DimCode, DimValueCode, OldDimSetID);

      TempDimensionSetEntry.DELETEALL;



      FOR i := 1 TO 20 DO

        IF (DimCode[i] <> '') AND (DimValueCode[i] <> '') THEN

          UPDATE(DimCode[i], DimValueCode[i]);

    END;



    PROCEDURE UPDATE@35(_DimCode@1001 : Code[20];_DimValueCode@1002 : Code[20]);

    BEGIN

      IF (_DimCode <> '') AND (_DimValueCode <> '') THEN BEGIN

        DELETE(_DimCode);

        CLEAR(TempDimensionSetEntry);

        TempDimensionSetEntry.VALIDATE("Dimension Set ID", -1);

        TempDimensionSetEntry.VALIDATE("Dimension Code", _DimCode);

        TempDimensionSetEntry.VALIDATE("Dimension Value Code", _DimValueCode);

        TempDimensionSetEntry.INSERT(TRUE);

      END;

    END;



    PROCEDURE DELETE@2(_DimCode@1001 : Code[20]);

    BEGIN

      IF TempDimensionSetEntry.GET(-1, _DimCode) THEN

        TempDimensionSetEntry.DELETE;

    END;



    PROCEDURE GetCurrDimValues@14(VAR _DimCode@1001 : ARRAY [20] OF Code[20];VAR _DimValueCode@1005 : ARRAY [20] OF Code[20];_DimensionSetID@1100074000 : Integer);

    VAR

      DimensionSetEntry@1100074001 : Record 480;

      GLSetupShortcutDimCode@1003 : ARRAY [8] OF Code[20];

      Filter@1006 : Text;

      i@1000 : Integer;

    BEGIN

      // Shortcut dimensions

      CLEAR(_DimCode);

      CLEAR(_DimValueCode);



      GetGLSetupShortcutDimCodes(_DimCode);

      FOR i := 1 TO 8 DO BEGIN

        IF _DimCode[i] <> '' THEN BEGIN

          IF Filter <> '' THEN

            Filter += '&';

          Filter += '<>' + _DimCode[i];

        END;



        IF DimensionSetEntry.GET(_DimensionSetID, _DimCode[i]) THEN

          _DimValueCode[i] := DimensionSetEntry."Dimension Value Code";

      END;



      // Other dimensions

      CLEAR(DimensionSetEntry);

      DimensionSetEntry.SETRANGE("Dimension Set ID", _DimensionSetID);

      DimensionSetEntry.SETFILTER("Dimension Code", Filter);

      IF DimensionSetEntry.FINDSET THEN

        REPEAT

          i += 1;

          _DimCode[i] := DimensionSetEntry."Dimension Code";

          _DimValueCode[i] := DimensionSetEntry."Dimension Value Code";

        UNTIL DimensionSetEntry.NEXT = 0;

    END;



    PROCEDURE GetDimValueCode@9(_DimCode@1000 : Code[20]) : Code[20];

    BEGIN

      IF TempDimensionSetEntry.GET(-1, _DimCode)THEN

        EXIT(TempDimensionSetEntry."Dimension Value Code");

    END;



    PROCEDURE GetDimSetID@88() : Integer;

    VAR

      DimensionManagement@1001 : Codeunit 408;

    BEGIN

      CLEAR(TempDimensionSetEntry);

      EXIT(DimensionManagement.GetDimensionSetID(TempDimensionSetEntry));

    END;



    PROCEDURE GetGLSetupShortcutDimCodes@15(VAR _DimCode@1001 : ARRAY [20] OF Code[20]);

    VAR

      GeneralLedgerSetup@1000 : Record 98;

    BEGIN

      CLEAR(_DimCode);

      GeneralLedgerSetup.GET;



      _DimCode[1] := GeneralLedgerSetup."Shortcut Dimension 1 Code";

      _DimCode[2] := GeneralLedgerSetup."Shortcut Dimension 2 Code";

      _DimCode[3] := GeneralLedgerSetup."Shortcut Dimension 3 Code";

      _DimCode[4] := GeneralLedgerSetup."Shortcut Dimension 4 Code";

      _DimCode[5] := GeneralLedgerSetup."Shortcut Dimension 5 Code";

      _DimCode[6] := GeneralLedgerSetup."Shortcut Dimension 6 Code";

      _DimCode[7] := GeneralLedgerSetup."Shortcut Dimension 7 Code";

      _DimCode[8] := GeneralLedgerSetup."Shortcut Dimension 8 Code";

    END;



    PROCEDURE CopyTempDimensionSetEntry@1100074015(VAR _TempDimensionSetEntry@1100074000 : TEMPORARY Record 480);

    BEGIN

      _TempDimensionSetEntry.DELETEALL;

      CLEAR(TempDimensionSetEntry);

      IF TempDimensionSetEntry.FINDSET THEN

        REPEAT

          _TempDimensionSetEntry.INIT;

          _TempDimensionSetEntry := TempDimensionSetEntry;

          _TempDimensionSetEntry.INSERT(TRUE);

        UNTIL TempDimensionSetEntry.NEXT = 0;

    END;



    BEGIN

    END.

  }

}



OBJECT Codeunit 92601 ExampleSDM

{

  OBJECT-PROPERTIES

  {

    Date=19/12/17;

    Time=12:00:00;

    Modified=Yes;

    Version List=SDM;

  }

  PROPERTIES

  {

    OnRun=BEGIN

            RunExample1;

            RunExample2;

          END;



  }

  CODE

  {



    LOCAL PROCEDURE RunExample1@1100074077();

    VAR

      TempSalesHeader@1100074004 : TEMPORARY Record 36;

      SpecDimensionManagement@1100074006 : Codeunit 92600;

      DimCode@1100074003 : ARRAY [20] OF Code[20];

      DimValueCode@1100074002 : ARRAY [20] OF Code[20];

      Originalvalues@1100074001 : Text;

      ModifiedValues@1100074005 : Text;

    BEGIN

      TempSalesHeader.INIT;

      TempSalesHeader."Document Type" := TempSalesHeader."Document Type"::Invoice;

      TempSalesHeader."No." := 'DUMMY_INV_NO';

      // Existing Dimension Set ID

      TempSalesHeader."Dimension Set ID" := 16;



      WITH SpecDimensionManagement DO BEGIN

        INIT(TempSalesHeader);

        GetCurrDimValues(DimCode, DimValueCode, TempSalesHeader."Dimension Set ID");



        Originalvalues := BuildMessage(TempSalesHeader, DimCode, DimValueCode);



        UPDATE('DEPARTEMENT', 'ADM');

        DELETE('REGION');

        TempSalesHeader."Dimension Set ID" := GetDimSetID;

        GetCurrDimValues(DimCode, DimValueCode, TempSalesHeader."Dimension Set ID");

      END;



      ModifiedValues := BuildMessage(TempSalesHeader, DimCode, DimValueCode);



      MESSAGE('%1\%2', Originalvalues, ModifiedValues);

    END;



    LOCAL PROCEDURE RunExample2@1100074012();

    VAR

      TempSalesHeader@1100074004 : TEMPORARY Record 36;

      SpecDimensionManagement@1100074006 : Codeunit 92600;

      DimCode@1100074003 : ARRAY [20] OF Code[20];

      DimValueCode@1100074002 : ARRAY [20] OF Code[20];

      Originalvalues@1100074001 : Text;

      ModifiedValues@1100074005 : Text;

    BEGIN

      TempSalesHeader.INIT;

      TempSalesHeader."Document Type" := TempSalesHeader."Document Type"::Invoice;

      TempSalesHeader."No." := 'DUMMY_INV_NO';

      // New Dimension Set ID

      // TempSalesHeader."Dimension Set ID" := 0;



      WITH SpecDimensionManagement DO BEGIN

        INIT(TempSalesHeader);

        GetCurrDimValues(DimCode, DimValueCode, TempSalesHeader."Dimension Set ID");



        Originalvalues := BuildMessage(TempSalesHeader, DimCode, DimValueCode);



        UPDATE('DEPARTEMENT', 'ADM');

        UPDATE('REGION', '40');

        TempSalesHeader."Dimension Set ID" := GetDimSetID;

        GetCurrDimValues(DimCode, DimValueCode, TempSalesHeader."Dimension Set ID");

      END;

      ModifiedValues := BuildMessage(TempSalesHeader, DimCode, DimValueCode);



      MESSAGE('%1\%2', Originalvalues, ModifiedValues);

    END;



    LOCAL PROCEDURE Template@1100074046();

    VAR

      MyRecord@1100074004 : TEMPORARY Record 36;

      SpecDimensionManagement@1100074006 : Codeunit 92600;

      DimCode@1100074003 : ARRAY [20] OF Code[20];

      DimValueCode@1100074002 : ARRAY [20] OF Code[20];

    BEGIN

      WITH SpecDimensionManagement DO BEGIN

        // Get Current Dimension Set ID or init a new one

        INIT(MyRecord);



        // Update my dimensions values

        UPDATE('DIM1', 'VALUE1');

        UPDATE('DIM2', 'VALUE2');

        UPDATE('DIM3', 'VALUE3');

        DELETE('DIM4');

        DELETE('DIM5');

        DELETE('DIM6');



        // Calculate my new Dimension Set ID

        MyRecord."Dimension Set ID" := GetDimSetID;



        // Get new values if needed

        GetCurrDimValues(DimCode, DimValueCode, MyRecord."Dimension Set ID");

      END;

    END;



    LOCAL PROCEDURE BuildMessage@1100074044(TempSalesHeader@1100074003 : TEMPORARY Record 36;DimCode@1100074002 : ARRAY [20] OF Code[20];DimCodeValue@1100074001 : ARRAY [20] OF Code[20]) Result : Text;

    VAR

      CrLf@1100074004 : Text[2];

      i@1100074005 : Integer;

    BEGIN

      CrLf[1] := 13;

      CrLf[2] := 10;



      Result := STRSUBSTNO('Values for "Dimension Set ID" = %1%2', TempSalesHeader."Dimension Set ID", CrLf);

      FOR i := 1 TO 20 DO

        IF (DimCode[i] <> '') AND (DimCodeValue[i] <> '') THEN

          Result += STRSUBSTNO('    %1 = %2%3', DimCode[i], DimCodeValue[i], CrLf);

    END;



    BEGIN

    END.

  }

}









